Computer Methods and Programs in Biomedicine 27 (1988) 79–82
Elsevier

CPB 00926

Section II. Systems and programs

ESTAGEST: an obstetrical application program in BASIC for the computation of estimated date of confinement

H. Frank Andersen and Mel L. Barclay

The Department of Obstetrics and Gynecology, The University of Michigan, Ann Arbor, MI, U.S.A.

The establishment of the estimated date of confinement (EDC) is an essential element in the care of pregnant women. Unfortunately, accurate information on the last menstrual period, the usual determinant of EDC, is not always available, or clinical estimations of gestational age may be discrepant. Prior studies have demonstrated that an average of several clinical examinations can provide a useful prediction of EDC. We have developed a simple BASIC computer program which allows the clinician to input up to 20 clinical examinations and determines the average EDC based on the clinical dates. The program displays the EDC for each individual examination, which provides an indication of the dispersion, or discrepancy, among the examinations. Average intervals to delivery date from the occurrence of clinical estimators of gestation are based on observed data at our institution and can be easily modified to fit local norms.

Obstetrics; Delivery date; Microcomputer; Date averaging; Prediction; Artificial intelligence

1. Introduction

One of the essential elements in the antenatal care of pregnant women is the establishment of their estimated date of delivery or confinement (EDC). The classical method of making this determination in clinical practice is Naegele's Rule. According to this method, one substracts three months and adds seven days to the date of onset of the last menstrual period (LMP). This method is rapid, easily performed, and provides an estimate which places the EDC 280 to 284 days from the date of the LMP.

There are, however, a large number of clinical situations in which the LMP is unknown or uncertain. Additionally, an accurately projected delivery date is particularly important in pregnancies complicated by medical problems such as maternal diabetes, pregnancy-induced hypertension, and erythroblastosis, all of which require well-timed delivery in order to insure maximum fetal survival.

2. Background

A number of clinical examinations have been used as estimators of gestational age in the human [1,2]. Among those estimators used frequently by the clinician are: (1) LMP, (2) appearance of fetal heart tones audible with an unamplified fetoscope, (3) uterine fundus at the pelvic brim or at the umbilicus, (4) sonographic examination of the fetus, and (5) quickening, the mother's first perception of the baby's movements.

Since all the methods of estimation involve some error, the combined information may be divergent and confusing [1–4]. However, the average of several clinical measures may be useful in predicting term delivery date [2]. Additionally, a

Correspondence: H.F. Andersen, Department of Obstetrics and Gynecology, L3221 Women's Hospital, Box 0264, The University of Michigan Medical Center, Ann Arbor, MI 48109-0264, U.S.A.

TABLE 1

Clinical examinations: mean interval to delivery in days [1]

Obstetrical milestone	*n*	Mean interval	S.D.
Last menstrual	297	284.2	14.6
Quickening in primipara	119	152.8	17.5
Quickening in multipara	85	161.4	17.6
Fetal heart audible	182	136.2	17.0
Uterus at umbilicus	95	140.8	14.9

measure of the statistical dispersion of the various clinical measurements may be useful in evaluating the array of clinical data.

The purposes of the program described here are twofold:

(1) To provide a computerized technique for establishment of a projected EDC given relevant clinical examinations.

(2) To provide information regarding the dispersion of projected delivery dates based on the various clinical examinations.

Each of the estimators specified above are associated with an average remaining duration of gestation, and thereby a projection of EDC. In work reported elswhere, constants have been developed which provide an EDC projection given the date of the occurrence of certain pregnancy-related events or measurements [1]. These constants are listed in Table 1.

3. Materials and methods

This program was originally developed in North Star BASIC, and has been modified for IBM PC BASIC. The source listing is compatible with most dialects of BASIC.

4. Algorithm

The predicted EDC from combined clinical estimators is obtained by calculating the average EDC based on the clinical estimators. Since each clinical estimator has a different level of accuracy for predicting the EDC a weighted average, allowing more influence for more accurate predictors would seem to be a preferable approach. However, earlier research has suggested that the simple average is as accurate in predicting the EDC as a weighted average [2].

The average EDC is calculated from the predicted intervals to delivery based on an arbitrary reference date. The computational tasks involved in this method are counting calendar days and calculating calendar dates.

There are numerous methodologies for the calculation of calendar dates and the intervals between two dates [5–8]. The Julian method, where days are numbered sequentially from a standard reference date (1 January 4713 B.C.), is not well-suited to microcomputers since the computations may occur at the limits of precision of 8- to 10-place integer arithmetic [6]. Although Julian techniques can be modified to operate on microcomputer systems, an alternative system was utilized. The system used here is based on the sequential day number in the year (modulo date). It was chosen because it presents the dates in a convenient way and circumvents problems with computational accuracy. A complete description of this methodology may be found in Andersen's paper [5]. Two subroutines in the program convert a date in month/day/year format into a modulo date and convert a modulo date back to month/day/year format. Specific subroutines convert calendar date into modulo date, modulo date back to calendar date, calculate the interval between two dates in modulo format, and add a specific interval in days to a modulo date.

5. Program description

The program is menu-driven and allows the user to choose any of a list of gestational estimators. In addition to gestational estimators with a specific, average interval to EDC (e.g., fetal heart tones first audible or quickening), the user may enter other gestational estimators (e.g., fundal height measurement) by entering the date of the measurement and the estimated gestational age by that measurement. After all of the relevant estimators are entered, the user chooses the final option,

'compute average EDC', which causes the printing of a summary table and the termination of the program.

Typing errors and inappropriate data entry must be corrected within the standard BASIC input procedures. The data must be entered in uniform fashion, that is, month, day, year form. The year may be entered as 19XX providing all subsequent entries are made in a similar fashion. This program does not provide for editing the list of data after entry.

Following data entry and assignment of variable names, the data are stored and converted from calendar-date form to modulo-date form. The estimated date of confinement associated with a particular date and clinical estimator is calculated by the addition of the appropriate number of days. The table of projected intervals from the gestational estimator to term delivery date can be modified according to local findings, if necessary. The interval from the reference date to the projected EDC is calculated and stored in the result matrix.

A sample output for a patient is shown in Fig. 1.

An annotated program listing is included in the appendix. The program is listed in structured format with variable definitions, data types, and constants specified. All subroutines are labelled.

```
Which of the following Gestational Estimators
do you wish to enter:

    Clinical Estimator                  Abbreviation   Average EGA

(1) Last Menstrual Period                    LMP    284 days to EDC
(2) Fetal Heart Tones Audible with Fetoscope FHT    20 weeks
(3) Uterine Fundus at Pelvic Brim            BRM    12 weeks
(4) Uterine Fundus at Umbilicus              UMB    20 weeks
(5) Quickening in a Primipara                QIKP   19 weeks
(6) Quickening in a Multipara                QIKM   17 weeks

(7) Ultrasound examination                   U/S    entry of EGA requested
(8) Any Other Gestational Estimator          OTH    entry of EGA requested

(0) Calculate the Projected Estimated Date of Confinement

Which estimator (enter number, please)? 1
Please enter the date of occurrence for this estimator.
  Month, Day, Year (separated by commas)? 10,1,86
     The E.D.C. for this estimator is:  7 / 12 / 87


(Menu is displayed again and data entry continues ...)
(Entry of an ultrasound examination is shown below:)


Which estimator (enter number, please)? 7
Please enter the date of occurrence for this estimator.
  Month, Day, Year (separated by commas)? 5,5,87
  How many weeks gestation by this estimator? 29
     The E.D.C. for this estimator is:  7 / 21 / 87


(Menu is displayed again and data entry continues ...)
(Calculation and display of results, after choosing '0':)


     THE OVERALL (average) PREDICTION OF E.D.C. is  7 / 9 / 87

                Table of Individual Estimators
       Date of Estimate     Estimator      Projected EDC
 1 .     10 / 1 / 86          LMP            7 / 12 / 87
 2 .     12 / 10 / 86         BRM            6 / 24 / 87
 3 .     2 / 24 / 87          UMB            7 / 15 / 87
 4 .     3 / 5 / 87           QIKP           8 / 5 / 87
 5 .     2 / 5 / 87           U/S            7 / 16 / 87
 6 .     3 / 30 / 87          OTH            6 / 15 / 87
 7 .     4 / 27 / 87          OTH            6 / 22 / 87
 8 .     5 / 5 / 87           U/S            7 / 21 / 87
```

Fig. 1. Computer output of a sample patient run. (User input is underlined.)

6. Discussion

A number of schemata are available for the statistical manipulation of EDC related data. Many relate to the issue of the uncertainty of projected EDCs [1–3]. In previous studies where the data was incomplete or conflicting, analytic methods using multiple linear regression techniques or graphic methodologies were found to be no more accurate than the simple arithmetic mean in the projection of EDC [2–4].

Computerized analysis of this type of calendar data also provides the clinician with a measurement of the dispersion of clinical estimators of gestational age. In this program we display predicted EDCs for each individual examination to provide the clinician with an indication of that dispersion. At present the significance of discrepancies among clinical estimators in relation to confidence intervals for predicting EDC is not known. Further research may allow calculation of confidence intervals for these estimates. This information may be factored into decisions regarding the value of a particular intervention as well as its relative risk.

References

[1] H.F. Andersen, T.R. Johnson Jr., M.L. Barclay and J.D. Flora Jr., Gestational age assessment. I. Analysis of individual clinical observations, Am. J. Obstet. Gynecol. 139 (1981) 173.

[2] H.F. Andersen, T.R. Johnson Jr., J.D. FLora Jr. and M.L. Barclay, Gestational age assessment. II. Prediction from combined clinical observations, Am. J. Obstet. Gynecol. 140 (1981) 770.

[3] T.R. Johnson Jr. and B.A. Work Jr., A dynamic graph for documentation of gestational age, Obstet. Gynecol. 54 (1979) 115.

[4] W.J. Ott, Accurate gestational dating, Obstet. Gynecol. 66 (1985) 311.

[5] H.F. Andersen, Calendar: a BASIC subroutine to manipulate calendar dates, Recreational Computing 8/3 (1979) 25.

[6] G. King, Julian dates for microcomputers, Dr. Dobbs J. 8/6 (June 1983) 66.

[7] R.G. Tantzen, The Collected Algorithms of the ACM (1963).

[8] S.G. Levy, BASIC dates (letter), Dr. Dobbs J. 9/1 (January 1984) 10.

Appendix

```
10 REM **********     ESTAGEST     **********
20 REM     Mel L. Barclay, M.D. and H. F. Andersen, M.D.
30 REM ***** ALL RIGHTS RESERVED, COPYRIGHT 1987 *****
40 REM             Ann Arbor, Michigan  48109
50 REM
60 REM Purpose of program:
70 REM  The calculation of estimated dates of delivery for human females.
80 REM  The program calculates calendar date when supplied with date of
90 REM  occurrence of various gestational estimators, and calculates the
100 REM average expected date of confinement.
130 REM
140 REM               *****     SYMBOLIC VARIABLES     *****
150 REM       M  - Month                      Y  - Year
160 REM       C  - Menu loop counter          D1 - Modulo date value
170 REM       D2 - Day counter                E  - Estimator number
180 REM       G  - Counter in table           I1 - Interval from reference date
190 REM       I9 - Average interval           J  - Month counter
200 REM       K  - Counter                    M1 - Days in the months, date
210 REM       R0 - Reference year             R3 - Decimal average EDC
220 REM       R4 - Intermediate sum
230 REM       W0 - Weeks EGA by U/S or ADD
240 REM       Z  - Number of days/yr          Z5 - Partial sum
250 REM       Z6 - Decimal value of I1        Z7 - Sum of intervals
270 REM
280 REM               *****     ARRAY VARIABLES     *****
290 REM       A$() - Array of estimator names
300 REM       E()  - The average number of days from estimator to delivery
310 REM                The data are base on research at the University of
320 REM                Michigan Medical Center: Andersen, H. F. , Johnson,
330 REM                T. R. B., Jr., Barclay, M. L., and Flora, J. D., Jr.
340 REM                Gestational age assessment I. Analysis of individual
350 REM                clinical observations.  Am J Obstet Gynecol 139:
360 REM                173-177, 1981.
370 REM       D()  - Table of dates entered    - Maximum of 20 entries allowed
380 REM       R()  - Table of computed results - Maximum of 20 entries allowed
390 REM
400 DIM A$(8), E(8), R(20,10), D(20,10)
410 REM     Set up some constants for time to parturition from
420 REM     various estimators.
430 E(1) = 284: A$(1) = "LMP" : REM LMP - Delivery
440 E(2) = 136: A$(2) = "FHT" : REM First Audible Fetal Heart Tones
450 E(3) = 196: A$(3) = "BRM" : REM Uterus at the pelvic brim
460 E(4) = 141: A$(4) = "UMB" : REM Uterus at the Umbilicus
470 E(5) = 153: A$(5) = "QIKP": REM Primip Quickening
480 E(6) = 161: A$(6) = "QIKM": REM Multip Quickening
490 E(7) = 0  : A$(7) = "U/S" : REM Ultrasound
500 E(8) = 0  : A$(8) = "OTH" : REM Other Estimator (such as fundal height)
510 DATA 31,28,31,30,31,30,31,31,30,31,30,31: REM  Data for CALENDAR subroutine
520 C = 0: REM Counter for number of estimators
530 REM DATA ENTRY ROUTINE BEGINS HERE
540 PRINT: PRINT: PRINT: PRINT: PRINT: PRINT
550 RESTORE
560 PRINT
570 PRINT "Which of the following Gestational Estimators"
580 PRINT "do you wish to enter:"
590 PRINT
600 PRINT "    Clinical Estimator                Abbreviation   Average EGA"
610 PRINT
620 PRINT "(1) Last Menstrual Period                    LMP   284 days to EDC"
630 PRINT "(2) Fetal Heart Tones Audible with Fetoscope  FHT   20 weeks"
640 PRINT "(3) Uterine Fundus at Pelvic Brim             BRM   12 weeks"
650 PRINT "(4) Uterine Fundus at Umbilicus               UMB   20 weeks"
660 PRINT "(5) Quickening in a Primipara                 QIKP  19 weeks"
670 PRINT "(6) Quickening in a Multipara                 QIKM  17 weeks"
680 PRINT
690 PRINT "(7) Ultrasound examination            U/S   entry of EGA requested"
700 PRINT "(8) Any Other Gestational Estimator   OTH   entry of EGA requested"
710 PRINT
720 PRINT "(0) Calculate the Projected Estimated Date of Confinement"
730 PRINT
740 INPUT "Which estimator (enter number, please)";E
750 IF (E  > 8) OR (E < 0) THEN 730
760 IF E = 0 THEN GOSUB 1310: REM Calculate average EDC
770 REM Data input
780     C = C+1: REM this is a counter for the number of estimators
790     PRINT "Please enter the date of occurrence for this estimator."
800     INPUT "  Month, Day, Year (separated by commas)";M,D,Y
810     IF C = 1 THEN R0 = Y: REM R0 is reference for calculating average EDC
820     D(C,1) = M: D(C,2) = D: D(C,3) = Y: REM Store data in array
830     IF E > =7 THEN GOSUB 930: REM Subroutine to get more info
840 GOSUB 980: REM Convert calendar date (M,D,Y) to modular form
850 GOSUB 1120: REM Calculate EDC for estimator
860 REM Print results for this estimator
870     PRINT "      The E.D.C. for this estimator is: ";M;"/";D;"/";Y
880     R(C,1) = C: R(C,2) = M: R(C,3) = D: R(C,4) = Y: REM  Store the results
890     R(C,5) = D1: R(C,6) = D1 - R0: R(C,7) = E:       REM    in R() array
900 GOTO 550: REM Continue the data entry loop until done
910 REM
920 REM SUBROUTINE to get more info on estimators 7 and 8
930      INPUT "  How many weeks gestation by this estimator";W0
940      E(E) = 280 - (W0 * 7): REM Calculate interval to predicted EDC
950 RETURN
960 REM
970 REM SUBROUTINE to convert date to YR.DAY (Modulo date)
980     RESTORE
990     D2 = 0
1000      FOR J = 1 TO 12
1010          IF J = M THEN 1070
1020           READ M1
1030           D2 = D2 + M1
1040           IF J <>  2 THEN 1060
1050           IF INT(Y/4) = Y/4 THEN D2 = D2 + 1
1060    NEXT J
1070    D1 = (D2 + D)/1000 + Y
1080    IF Y/4 = INT(Y/4) THEN Z = 366 ELSE Z = 365
1090 RETURN
1100 REM
1110 REM SUBROUTINE to figure EDC on basis of estimator
1120     D1 = D1 + E(E)/1000
1130 REM  Convert D1 (predicted EDC back to calendar date)
1140     IF (D1 - INT(D1))  > .365 THEN D1 = D1 + .635
1150     RESTORE
1160     Y = INT(D1)
1170     D = 1000*(D1 - Y)
1180     M = 0
1190     FOR K = 1 TO 12
1200          M = M + 1
1210          READ M1
1220          IF M <>2 THEN 1240
1230          IF INT(Y/4) = Y/4 THEN M1 = M1 + 1
1240          IF D >  M1 THEN D = D - M1 ELSE 1260
1250     NEXT K
1260     IF INT(D) = 0 THEN D = 1
1270     D = INT(D + .5)
1280 RETURN
1290 REM
1300 REM SUBROUTINE to calculate average EDC and print results
1310     FOR K = 1 TO C
1320          Z5 = (R(K,6) - INT(R(K,6))) * 1000/Z
1330          Z6 = INT(R(K,6)) + Z5
1340          Z7 = Z7 + Z6
1360     NEXT K
1370 REM  Calculate the average decimal interval, I9
1380     I9 = Z7/C
1410     R3 = R0 + I9
1420     R4 = (R3 - INT(R3)) * Z/1000
1430     D1 = R4 + INT(R3)
1440     GOSUB 1140                              : REM Calculate final EDC
1450 REM Print results in a table
1460 PRINT: PRINT: PRINT: PRINT: PRINT: PRINT
1470 PRINT "      THE OVERALL (average) PREDICTION OF E.D.C. is ";M;"/";D;"/";Y
1480 PRINT "-------------------------------------------------------------"
1490 PRINT "               Table of Individual Estimators"
1500 PRINT "       Date of Estimate    Estimator       Projected EDC"
1510 PRINT "-------------------------------------------------------------"
1520 FOR K = 1 TO C
1530     PRINT R(K,1);".    ";D(K,1);"/";D(K,2);"/";D(K,3),
1540     PRINT A$(R(K,7)),R(K,2);"/";R(K,3);"/";R(K,4)
1550 NEXT K
1560 PRINT "-------------------------------------------------------------"
1610 END
```